
Timing Considerations with VHDL-Based Designs

This tutorial describes how Altera’s QuartusR© II software deals with the timing issues in designs based on the
VHDL hardware description language. It discusses the various timing parameters and explains how specific tim-
ing constraints may be set by the user.

Quartus II software includes a Timing Analyzer module whichperforms a detailed analysis of all timing delays
for a circuit that is compiled for implementation in an FPGA chip. This tutorial discusses the types of analyses
performed and shows how particular timing requirements maybe specified by the user. The discussion assumes
that the reader is familiar with the basic operation of Quartus II software, as may be learned from an introductory
tutorial.

Doing this tutorial, the reader will learn about:

• Parameters evaluated by the Timing Analyzer

• Specifying the desired values of timing parameters

• Using timing simulation

The timing results shown in the examples in this tutorial were obtained using Quartus II version 5.0, but other
versions of the software can also be used.

1 Example Circuit

Timing issues are most important in circuits that involve long paths through combinational logic elements with
registers at inputs and outputs of these paths. As an example, we will use the adder/subtractor circuit shown in
Figure 1. It can add, subtract, and accumulaten-bit numbers using the 2’s complement number representation.
The two primary inputs are numbersA = an−1an−2 · · · a0 andB = bn−1bn−2 · · · b0, and the primary output is
Z = zn−1zn−2 · · · z0. Another input is theAddSubcontrol signal which causesZ = A+B to be performed when
AddSub= 0 andZ = A − B whenAddSub= 1. A second control input,Sel, is used to select the accumulator
mode of operation. IfSel= 0, the operationZ = A ± B is performed, but ifSel= 1, thenB is added to or
subtracted from the current value ofZ. If the addition or subtraction operations result in arithmetic overflow, an
output signal,Overflow, is asserted.

To make it easier to deal with asynchronous input signals, they are loaded into flip-flops on a positive edge of
the clock. Thus, inputsA andB will be loaded into registersAregandBreg, while SelandAddSubwill be loaded
into flip-flopsSelRandAddSubR, respectively. The adder/subtractor circuit places the result into registerZreg.

Figure 1. The adder/subtractor circuit.

The required circuit is described by the VHDL code in Figure 2. For our example, we use a 16-bit circuit as
specified byn = 16. Implement this circuit as follows:

• Create a projectaddersubtractor.

• Include a fileaddersubtractor.vhd, which corresponds to Figure 2, in the project. For convenience, this file is
provided in the directoryDE2_tutorials\design_files, which is included on the CD-ROM that accompanies
the DE2 board and can also be found on Altera’s DE2 web pages.

• Choose the Cyclone II EP2C35F672C6 device, which is the FPGAchip on Altera’s DE2 board.

• Compile the design.

LIBRARY ieee ;
USE ieee.std_logic_1164.all ;

−− Top-level entity
ENTITY addersubtractor IS

GENERIC ( n : INTEGER := 16 ) ;
PORT (A, B : IN STD_LOGIC_VECTOR(n−1 DOWNTO 0) ;

Clock, Reset, Sel, AddSub : IN STD_LOGIC ;
Z : BUFFER STD_LOGIC_VECTOR(n−1 DOWNTO 0) ;
Overflow : OUT STD_LOGIC ) ;

END addersubtractor ;

ARCHITECTURE Behavior OF addersubtractor IS
SIGNAL G, H, M, Areg, Breg, Zreg, AddSubR_n : STD_LOGIC_VECTOR(n-1 DOWNTO 0) ;
SIGNAL SelR, AddSubR, carryout, over_flow : STD_LOGIC ;
COMPONENT mux2to1

GENERIC ( k : INTEGER := 8 ) ;
PORT ( V, W : IN STD_LOGIC_VECTOR(k-1 DOWNTO 0) ;

Selm : IN STD_LOGIC ;
F : OUT STD_LOGIC_VECTOR(k-1 DOWNTO 0) ) ;

END COMPONENT ;
COMPONENT adderk

GENERIC ( k : INTEGER := 8 ) ;
PORT (carryin : IN STD_LOGIC ;

X, Y : IN STD_LOGIC_VECTOR(k-1 DOWNTO 0) ;
S : OUT STD_LOGIC_VECTOR(k-1 DOWNTO 0) ;
carryout : OUT STD_LOGIC ) ;

END COMPONENT ;
BEGIN

PROCESS ( Reset, Clock )
BEGIN

IF Reset = ’1’ THEN
Areg<= (OTHERS => ’0’); Breg <= (OTHERS => ’0’);
Zreg<= (OTHERS => ’0’); SelR <= ’0’; AddSubR<= ’0’; Overflow <= ’0’;

ELSIF Clock’EVENT AND Clock = ’1’ THEN
Areg<= A; Breg<= B; Zreg<= M;
SelR<= Sel; AddSubR<= AddSub; Overflow<= over_flow;

END IF ;
END PROCESS ;

nbit_adder: adderk
GENERIC MAP ( k => n )
PORT MAP ( AddSubR, G, H, M, carryout ) ;

multiplexer: mux2to1
GENERIC MAP ( k => n )
PORT MAP ( Areg, Z, SelR, G ) ;

AddSubR_n<= (OTHERS => AddSubR) ;
H <= Breg XOR AddSubR_n ;
over_flow<= carryout XOR G(n−1) XOR H(n−1) XOR M(n−1) ;
Z <= Zreg ;

END Behavior;
. . . continued in Partb

Figure 2. VHDL code for the circuit in Figure 1 (Parta).

−− k-bit 2-to-1 multiplexer
LIBRARY ieee ;
USE ieee.std_logic_1164.all ;

ENTITY mux2to1 IS
GENERIC ( k : INTEGER := 8 ) ;
PORT ( V, W : IN STD_LOGIC_VECTOR(k−1 DOWNTO 0) ;

Selm : IN STD_LOGIC ;
F : OUT STD_LOGIC_VECTOR(k−1 DOWNTO 0) ) ;

END mux2to1 ;

ARCHITECTURE Behavior OF mux2to1 IS
BEGIN

PROCESS ( V, W, Selm )
BEGIN

IF Selm = ’0’ THEN
F <= V ;

ELSE
F <= W ;

END IF ;
END PROCESS ;

END Behavior ;

−− k-bit adder
LIBRARY ieee ;
USE ieee.std_logic_1164.all ;
USE ieee.std_logic_signed.all ;

ENTITY adderk IS
GENERIC ( k : INTEGER := 8 ) ;
PORT ( carryin : IN STD_LOGIC ;

X, Y : IN STD_LOGIC_VECTOR(k−1 DOWNTO 0) ;
S : OUT STD_LOGIC_VECTOR(k−1 DOWNTO 0) ;
carryout : OUT STD_LOGIC ) ;

END adderk ;

ARCHITECTURE Behavior OF adderk IS
SIGNAL Sum : STD_LOGIC_VECTOR(k DOWNTO 0) ;

BEGIN
Sum<= (’0’ & X) + (’0’ & Y) + carryin ;
S<= Sum(k−1 DOWNTO 0) ;
carryout<= Sum(k) ;

END Behavior ;

Figure 2. VHDL code for the circuit in Figure 1 (Partb).

2 Timing Analyzer Report

Successful compilation of our circuit generates the Compilation Report in Figure 3. This report provides a lot
of useful information. It shows the number of logic elements, flip-flops (called registers), and pins needed to
implement the circuit. It gives detailed information produced by the Synthesis and Fitter modules. It also indicates
the speed of the implemented circuit. A good measure of the speed is the maximum frequency at which the circuit
can be clocked, referred to asfmax. This measure depends on the longest delay along any path, called thecritical
path, between two registers clocked by the same clock. Quartus IIsoftware performs a timing analysis to determine
the expected performance of the circuit. It evaluates several parameters, which are listed in the Timing Analyzer
section of the Compilation Report. Click on the small + symbol next to Timing Analyzer to expand this section of
the report, and then click on the Timing Analyzer itemSummary which displays the table in Figure 4. The last
entry in the table shows that the maximum frequency for our circuit implemented on the specified chip is 214.27
MHz. You may get a different value offmax, dependent on the specific version of Quartus II software that you are
using. To see the paths in the circuit that limit thefmax, click on the Timing Analyzer itemClock Setup: ’Clock’
in Figure 4 to obtain the display in Figure 5. This table showsthat the critical path begins at the flip-flopAddSubR
and ends at the flip-flopOverflow.

Figure 3. The Compilation Report.

Figure 4. The Timing Analyzer Summary.

Figure 5. Critical paths.

The table in Figure 4 also shows other timing results. Whilefmaxis a function of the longest propagation delay
between two registers in the circuit, it does not indicate the delays with which output signals appear at the pins of
the chip. The time elapsed from an active edge of the clock signal at the clock source until a corresponding output
signal is produced (from a flip-flop) at an output pin is denoted as thetco delay at that pin. In the worst case, the
tco in our circuit is 7.750 ns. Click ontco in the Timing Analyzer section to view the table given in Figure 6. The
first entry in the table shows that it takes 7.750 ns from when an active clock edge occurs until a signal propagates
from bit 8 in registerZregto the output pinz8. The other two parameters given in Figure 4 are setup time,tsu, and
hold time,th.

Figure 6. Thetcodelays.

3 Specifying Timing Constraints

So far we have compiled our VHDL code without indicating to the Quartus II software the required speed per-
formance of the circuit. In the absence of such timing constraints the Quartus II software implements a designed
circuit in a good but not necessarily the best way in order to keep the compilation time short. If the result does not
meet the user’s expectations, it is possible to specify certain timing constraints that should be met. For example,
suppose that we want our example circuit to operate at a clockfrequency of at least 250 MHz, rather than the
214.27 MHz as indicated by the value offmax in Figure 4. To see if this can be achieved we can set thefmax
constraint as follows:

1. SelectAssignments > Timing Settings to reach theTiming Requirements & Options window in Figure
7. In this window it is possible to specify the requirements for a number of different parameters.

2. In the box Clock Settings specify that the required value of fmaxis 250 MHz. ClickOK.

3. Recompile the circuit.

4. Open the Timing Analyzer Summary to see that the newfmaxis 263.02 MHz, as indicated in Figure 8. You
may get a slghtly different result depending on the version of the Quartus II software used.

Figure 7. Specify the timing constraints in the Settings window.

Figure 8. New timing results.

If the specified constraint is too high, the Quartus II compiler will not be able to satisfy it. For example, set the
fmaxconstraint to 300 MHz and recompile the circuit. Now, the Timing Analyzer Summary will show that this
constraint cannot be met, as seen in Figure 9.

Figure 9. The timing constraint cannot be met.

The specifiedfmaxof 300 MHz cannot be achieved because one or more paths in the circuit have long propa-
gation delays. To locate the most critical path highlight the Clock Setup entry in the table by clicking on it. Then,
right-click to get the pop-up menu shown in Figure 10. SelectLocate > Locate in RTL Viewer which will
cause the RTL Viewer to display the critical path as presented in Figure 11. Note that this path begins at flip-flop
AddSubRand ends at theOverflowflip-flop.

Figure 10. Locate the critical path.

Figure 11. Path for which the timing constraint cannot be met.

It is likely that there are other paths that make it impossible to meet the specified constraint. To identify these
paths chooseClock Setup: ’Clock’ on the left side of the Compilation Report in Figure 9. As seenin Figure 12,
there are 10 paths with propagation delays that are too long.Observe a column labeled Slack. The termslack is
used to indicate the margin by which a timing requirement is met or not met. In the top row in Figure 12 we see
that the timing delays along the path from theAddSubRflip-flop to theOverflowflip-flop are 0.469 ns longer than
the maximum of 4 ns that is the period of the 250-MHz clock specified as thefmaxconstraint.

Figure 12. The longest delay paths.

We have shown how to set thefmaxconstraint. The other constraints depicted in the window inFigure 7 can
be set in the same way.

4 Timing Simulation

Timing simulation provides a graphical indication of the delays in the implemented circuit, as can be observed
from the displayed waveforms. For a discussion of simulation see the tutorialQuartus II Simulation with VHDL
Designs, which uses the sameaddersubtractorcircuit as an example.

